

Dodge the Creeps!

El joc es diu **Dodge the Creeps!**. El teu personatge ha de moure's i evitar els enemics tant de temps com sigui possible.


Continuació de la fitxa anterior!

55. Ara unirem tot el que hem fet en una escena de joc jugable.

Creeu una escena nova i afegiu un **Node** anomenat **Main**.

La raó per la qual estem utilitzant **Node** en lloc de **Node2D** és perquè aquest node serà un contenidor per a gestionar la lògica del joc. No requereix funcionalitats 2D.

56. Feu clic al botó **Instance** i seleccioneu **Player.tscn**.



57. Ara, afegiu els següents nodes com a fills del **Main**, i poseu-los un nom com es mostra:

- Un **Timer** anomenat **MobTimer** per a controlar la freqüència de generació dels **Mobs**.
- Un **Timer** anomenat **ScoreTimer** per a incrementar la puntuació cada segon.
- Un **Timer** anomenat **StartTimer** per a crear un retard abans d'iniciar.
- Un **Position2D** nomenat **StartPosition** per indicar la posició inicial del jugador.

Estableix la propietat **Wait Time** de cadascun dels nodes **Timer** de la manera següent, tenint en compte que els valors són en segons.

- **MobTimer**: 0.5
- **ScoreTimer**: 1
- **StartTimer**: 2

58. Establiu la propietat **One Shot** de **StartTimer** a **On** i establiu la propietat **Position** del node **StartPosition** a (240, 450).

[COMMIT] Escena Principal.

59. El node **Main** generarà noves **Mobs**, i volem que apareguin en una ubicació aleatòria a la vora de la pantalla.

Afegeix un node **Path2D** anomenat **MobPath** com a fill de **Main**. Quan seleccioneu **Path2D**, veureu alguns botons nous a la part superior de l'editor:



60. Seleccioneu el del mig, **Add Point** i dibuixeu el camí fent clic per afegir els punts a les cantonades que es mostren.

Per a fer que els punts s'ajustin a la quadrícula, assegureu-vos que tant **Use Grid Snap** com **Use Snap** estan seleccionats.

Aquestes opcions es poden trobar a l'esquerra del botó **Lock**.




61. Dibuixa el camí en sentit horari, o les teves **Mobs** generaran apuntant cap a fora en comptes d'endins.



62. Després de col·locar el punt **4** a la imatge, feu clic al botó **Close Curve** i la corba es completarà.

63. Ara que el camí està definit, afegiu un node **PathFollow2D** com a fill de **MobPath** i poseu-li el nom **MobSpawnLocation**.

Aquest node girarà automàticament i seguirà el camí a mesura que es mou, de manera que podem utilitzar-lo per a seleccionar una posició i direcció aleatòries al llarg del camí.



[COMMIT] Generació de Mobs.

64. Afegeix un script al **Main**. A la part superior de l'script, utilitzem **export (PackedScene)** per a permetre'ns triar l'escena **Mob** que volem instanciar.

```
extends Node
```

```
export(PackedScene) var mob_scene
var score
```

65. També s'afegeix una crida a **randomize()** perquè el generador de nombres aleatoris generi diferents nombres aleatoris cada vegada que s'executa el joc:

```
func _ready():
    randomize()
```

66. Feu clic al node **Main** i veureu la propietat **Mob Scene** al menú, sota **Script Variables**.

Podeu assignar el valor d'aquesta propietat de dues maneres:

- Arrossegueu **Mob.tscn** des del menú **FileSystem** i deixeu-lo anar a la propietat **Mob Scene**.
- Feu clic a la fletxa cap avall al costat de **[empty]** i trieu **Load**. Seleccioneu **Mob.tscn**.

67. A continuació, seleccioneu el node **Player** al menú **Scene** i accediu al menú **Node** a la barra lateral. Assegureu-vos de tenir la pestanya **Senyals** seleccionada al menú **Node**.

68. Hauríeu de veure una llista dels senyals del node **Player**. Cerqueu i feu doble clic al senyal **hit** (o feu-hi clic amb el botó dret i seleccioneu **Connect...**). Això obrirà el diàleg de connexió del senyal.

69. Volem fer una nova funció anomenada **game_over**, que gestionarà el que ha de passar quan acabi un joc.

Escriviu **game_over** al quadre **Receiver Method** a la part inferior del diàleg de connexió de senyal i feu clic a **Connect**.

Afegeix el codi següent a la nova funció, així com una funció **new_game** que ho configurarà tot per a un joc nou:

```
func game_over():
    $ScoreTimer.stop()
    $MobTimer.stop()

func new_game():
    score = 0
    $Player.start($StartPosition.position)
    $StartTimer.start()
```

70. Connecteu ara el senyal **timeout()** de cadascun dels nodes **Timer** (**StartTimer**, **ScoreTimer** i **MobTimer**) a l'script principal.

StartTimer iniciarà els altres dos **Timers** i **ScoreTimer** incrementarà la puntuació.

```
func _on_ScoreTimer_timeout():
    score += 1

func _on_StartTimer_timeout():
    $MobTimer.start()
    $ScoreTimer.start()
```

71. A `_on_MobTimer_timeout()`, crearem una instància de **Mob**, triarem una ubicació inicial aleatòria al llarg del **Path2D** i posarem la **Mob** en moviment.

El node **PathFollow2D** girarà automàticament a mesura que segueixi el camí, de manera que l'utilitzarem per a seleccionar la direcció de la **Mob** així com la seva posició.

Quan generem una **Mob**, triarem un valor aleatori entre **150.0** i **250.0** per la rapidesa amb què es mourà cada **Mob** (seria avorrit si totes es moguessin a la mateixa velocitat).

Tingueu en compte que s'ha d'afegir una nova instància a l'escena utilitzant `add_child()`.

```
func _on_MobTimer_timeout():
    # Crea una nova instància de l'escena Mob.
    var mob = mob_scene.instance()

    # Trieu una ubicació aleatòria a Path2D.
    var mob_spawn_location = get_node("MobPath/MobSpawnLocation")
    mob_spawn_location.offset = randi()

    # Estableix la direcció de la multitud perpendicular a la direcció del camí.
    var direction = mob_spawn_location.rotation + PI / 2

    # Estableix la posició de la multitud a una ubicació aleatòria.
    mob.position = mob_spawn_location.position

    # Afegeix una mica d'aleatorietat a la direcció.
    direction += rand_range(-PI / 4, PI / 4)
    mob.rotation = direction

    # Tria la velocitat de la multitud.
    var velocity = Vector2(rand_range(150.0, 250.0), 0.0)
    mob.linear_velocity = velocity.rotated(direction)

    # Va fer que la multitud s'afegís a l'escena principal.
    add_child(mob)
```

[COMMIT] Main Script.

72. Posem a prova l'escena per assegurar-nos que tot funciona. Afegeix aquesta crida ***new_game*** a ***_ready()***:

```
func _ready():  
    randomize()  
    new_game()
```

73. Assignem també ***Main*** com la nostra ***Main Scene***, és a dir, la que s'executa automàticament quan es llança el joc. Premeu el botó ***Play*** i seleccioneu ***Main.tscn*** quan se us demani.

74. Hauries de ser capaç de moure el jugador, veure les ***Mobs*** generant-se i veure que el jugador desapareix quan xoca amb una ***Mob***.

75. Feu clic al botó ***Instance*** i seleccioneu ***Player.tscn***.

76. Quan esteu segur que tot funciona, suprimiu la crida a ***new_game()*** de ***_ready()***.